複素数と Mathematica

桂田 祐史

2015 年 9 月 28 日, 2020 年 1 月 19 日

http://nalab.mind.meiji.ac.jp/~mk/complex/mathematica-memo/

講義の前半は割とマメに Mathematica の例を示しているけれど、後半はさぼっているので、今年度こそ…

実際に Mathematica で試してもらうことを前提に、結果がどうなるかは省略してあるのが多い。悪しからず…(某年某月某日加筆: Mathematica のバージョンによって挙動が違うことがあるので、結果も載せた方が良いのかもしれない。)

(2019/11/20) 久しぶりに見直して、内容がイマイチであることに気づいた。ここに書いていないことをいくつか、http://nalab.mind.meiji.ac.jp/~mk/lecture/complex-function-2019/20191120_complex_function.pdf の中でやっている。こちらを更新すべきであるが、今猛烈に忙しいので、なかなか手をつけられない。

(2020/1/19) 学生で「Wolfram Alpha[1] でやっています」という人もいる。確かに、ここに載っていることくらいは、Wolfram Alpha でもやってくれるみたいだ。

1 四則など簡単な演算

虚数単位は `I` (大文字) で表す。絶対値 (absolute value) は `Abs[]`, 偏角 (argument) の主値は `Arg[]`, 共役複素数 (complex conjugate) は `Conjugate[]` で計算できる。

問 1[2] (1) の検算に使ってみる。

```
a=1+I
b=2+3I
a+b
a-b
a b
a/b
Abs[a]
Conjugate[a]
Arg[a]
a^4
```

[1] https://www.wolframalpha.com/

[2] http://nalab.mind.meiji.ac.jp/~mk/lecture/complex-function-2015/toi1.pdf

2　平方根の計算

$c = a + ib$ $(a, b \in \mathbb{R})$ が与えられた時に、$z^2 = c$ の解を $z = x + iy$ $(x, y \in \mathbb{R})$ の形で求めることが出来る。
(複素数の平方根が、実数の $\sqrt{\quad}$ で表現できる、という定理に基づく。)
$(x + iy)^2 = a + ib$ より連立方程式

$$x^2 - y^2 = a, \quad 2xy = b$$

が得られるので、その実数の範囲の解を求めれば良い。

$1 + i$ の平方根を求めてみよう。

```
a=1
b=1
sol=Solve[{x^2-y^2==a,2 x y==b},{x,y},Reals]
FullSimplify[sol]
```

以上は授業で説明したやり方に沿って Mathematica に仕事をさせるものだが、z の方程式のまま解かせることも出来る (Mathematica が内部で何をしているのかは謎だけど)。

```
sol=Solve[z^2 == 1 + I, z]
sol2=ComplexExpand[sol]
ToRadicals[sol2]
```

最初に $z^2 = 1 + i$ を解かせると $z = \pm\sqrt{1+i}$ となるが、`ComplexExpand[]` で実部・虚部に展開させると、$z = \pm\sqrt[4]{2}\left(\cos\frac{\pi}{8} + i\sin\frac{\pi}{8}\right)$ となり、`ToRadicals[]` で処理すると、$z = \pm\left(\frac{\sqrt{2+\sqrt{2}}}{2^{3/4}} + i\frac{\sqrt{2-\sqrt{2}}}{2^{3/4}}\right)$ となる。

3　実部虚部への分解

$\Omega \subset \mathbb{C}$, $f\colon \Omega \to \mathbb{C}$ とするとき、

$$u(x, y) := \operatorname{Re} f(x + iy), \quad v(x, y) := \operatorname{Im} f(x + iy) \quad \left((x, y) \in \widetilde{\Omega} := \left\{(x, y) \in \mathbb{R}^2 \mid x + iy \in \Omega\right\}\right)$$

で定めた u, v (複素関数の実部・虚部) が必要になる場合がある。

これらを求めるには、既に紹介した `ComplexExpand[]` を用いると良い。

```
f[z_]:=z^3
ComplexExpand[f[x+I y]]

g[z_]:=Cos[z]
ComplexExpand[g[x+I y]]
```

4　(有限項) Taylor 展開

有限項までで打ち切ったものしか計算出来ないが、`Series[式, {変数, c, n}]` で、c のまわりの Taylor 展開を第 n 項まで計算出来る。

```
Series[Cos[z],{z,0,10}]

Series[1/(2+z),{z,0,10}]
```

5 級数

Mathematica の Sum[] は級数の和を計算してくれる。思いの外、強力である。

```
Sum[(-1)^n z^n/(2n)!,{n,0,Infinity}]
```

に対して Cos[z] という答を返す。

Taylor 展開、Laurent 展開は、ある程度、自分で計算する必要があるようだが、その結果を Sum[] に与えることで検算が可能である。

6 部分分数分解

有理式の部分分数への分解が必要になる場合があるが、Apart[] で計算できる。

```
Apart[(z^3-3z^2-z+5)/(z^2-5z+6)]
```

7 応用: ある計算問題 (有理関数の Taylor 展開を求める) の答の検算

$$f(z)=\frac{z^3-3z^2-z+5}{z^2-5z+6}.$$

の 0 の周りの Taylor 展開を求めよ、という問題。

人手で解く場合は、$f(z)$ を部分分数に分解する。そのためには、z^3-3z^2-z+5 を z^2-5z+6 で割りたくなる。

```
A=z^3-3z^2-z+5
B=z^2-5z+6
f=A/B
q=PolynomialQuotient[A,B,z]
r=PolynomialRemainder[A,B,z]
```

これから商 $q=z+2$, 余り $r=3z-7$ が求まる (商は quotient, 余りは remainder. polynomial は多項式という意味)。ゆえに

$$f(z)=\frac{(z+2)(z^2-5z+6)+3z-7}{z^2-5z+6}=z+2+\frac{3z-7}{(z-2)(z-3)}.$$

この右辺を部分分数分解しても良いが、そもそも Mathematica にやらせるのならば、最初から $f(z)$ の部分分数分解を指示してもよい。

```
Apart[r/B]

Apart[f]
```

それぞれ $\dfrac{2}{-3+z}+\dfrac{1}{-2+z}$, $2+\dfrac{2}{-3+z}+\dfrac{1}{-2+z}+z$ となる。

結局

$$f(z)=z+2+\frac{1}{z-2}+\frac{2}{z-3}.$$

Series[f,{z,0,10}] とすると、0 の周りの Taylor 展開を 10 次の項まで求めることが出来る。

$$f(z)=\frac{5}{6}+\frac{19z}{36}-\frac{43}{216}z^2-\frac{113}{1296}z^3-\frac{307}{7776}z^4-\cdots(\text{途中省略})-\cdots-\frac{181243}{362797056}z^{10}+O(z^{11})$$

が得られる。series は級数という意味の英単語である。

残念ながら、Mathematica で Taylor 展開の一般項を求めるための簡単な手段は用意されていないようである。

人手で計算して、

$$f(z)=\frac{5}{6}+\frac{19}{36}z-\sum_{n=2}^{\infty}\left(\frac{1}{2^{n+1}}+\frac{2}{3^{n+1}}\right)z^n$$

が求まるが、この結果の検算は可能で

```
5/6+19z/36-Sum[(1/2^(n+1)+2/3^(n+1))z^n,{n,2,Infinity}]
Simplify[%]
```

とすると、

$$\frac{5-z-3z^2+z^3}{6-5z+z^2}$$

が得られる。無事、$f(z)$ と一致したので、ほっと一息。

8 複素対数関数を描く

2 変数 (x,y) の関数としての $\operatorname{Im}\operatorname{Log}(x+iy)$, $\operatorname{Re}\operatorname{Log}(x+iy)$ のグラフを描いてみよう。

それぞれ $\operatorname{Arg}(x+iy)$, $\log\sqrt{x^2+y^2}$ であるから、コンピューターで図示しなくても分からなくはないが (図示しなくても分かるけれど)、やってみることを勧める。

Plot3D[] や ContourPlot[] では、描画範囲を x 座標と y 座標の範囲で指定するので、変数は x+I y と書くと良い (小さなノウハウ)。

```
Plot3D[Boole[x^2+y^2<4] Im[Log[x+I y]],{x,-2,2},{y,-2,2}]

Plot3D[Boole[x^2+y^2<4] Re[Log[x+I y]],{x,-2,2},{y,-2,2}]
```




(Boole[] はなくても良い。描画範囲を円 $x^2+y^2=4$ 内に制限しているだけで、あまり深い意味はない。個人的な趣味です。)

Mathematica で描いたグラフは、マウスでつかんでグリグリ動かせる。ぜひやってみること (静止画を見るだけだと今ひとつ分かりにくい)。

Plot3D[] の代わりに ContourPlot[] を用いると、レベル表示 (≒等高線描画) 出来る。

```
ContourPlot[Im[Log[x+I y]], {x,-2,2}, {y,-2,2}, Contours->Table[x,{x,-Pi,Pi,Pi/8}]]

ContourPlot[Re[Log[x+I y]],{x,-2,2},{y,-2,2}]
```

9　Abel の連続性定理に現れる収束範囲

冪級数 $\displaystyle\sum_{n=0}^{\infty} a_n(z-c)^n$ が収束円周上のある点 z_0 で収束するならば、その冪級数は "**Stolz の角領域**" で一様収束するので、和はそこで連続な関数である、というのが Abel の連続性定理で、それにより、

$$\frac{\pi}{4}=1-\frac{1}{3}+\frac{1}{5}-\cdots+\frac{(-1)^{n-1}}{2n-1}+\cdots$$

などの有名な結果が証明できる。

例えば $c=0$, $z_0=R$ の場合、

$$\Omega_K:=\left\{z\in\mathbb{C}\,\middle|\,|z|<R,\quad \frac{|1-z/R|}{1-|z|/R}<K\right\}$$

であるが、これは一体どういう形なのだろうか？

```
stolz[K_, R_] :=
 Block[{g1, g2},
  g1 = ContourPlot[x^2 + y^2 == R^2, {x, -2 R, 2 R}, {y, -2 R, 2 R}];
  g2 = RegionPlot[
    x^2 + y^2 < R^2 &&
     Abs[1 - (x + I y)/R]/(1 - Abs[x + I y]/R) <= K, {x, -2 R,
     2 R}, {y, -2 R, 2 R}]; Show[g1, g2]
 ]

R=1
Manipulate[stolz[K,R],{K,1,10,0.2}]
```



図 1: K を大きくすると膨れていきます

筆者は、Mathematica を使うまで、Ω_K がどういう形をしているか、実は良く分かっていなかった (そんなに難しくもないけれど、ちょっと考えて分かるものでもなくて、何となく気になってはいたけれど、放置していました。)。

10 線積分の験算

以下の線積分の値を求めよ。

(1) $C\colon z = t + it^2$ $(t \in [0,1])$ とするとき $I_1 = \displaystyle\int_C \operatorname{Re} z\, dz$ (2) $c \in \mathbb{C}$, $r > 0$, $n \in \mathbb{N}$, $C\colon z = c + re^{i\theta}$ $(\theta \in [0, 2\pi])$ とするとき $I_2 = \displaystyle\int_C \frac{dz}{(z-c)^n}$ (3) 0 から $1+i$ に至る線分を C とするとき $I_3 = \displaystyle\int_C \operatorname{Im} z\, dz$ (4) 単位円 $|z| = 1$ の下半分を -1 から 1 までたどる曲線を C とするとき $I_4 = \displaystyle\int_C \overline{z}\, dz$ (5) 図の正方形の周を反時計回りに一周する曲線を C とするとき $I_5 = \displaystyle\int_C |z|\, dz$, $I_6 = \displaystyle\int_C \left(z^2 + 3z + 4\right) dz$



```
z=t+I t^2;
I1=Integrate[Re[z]D[z,t],{t,0,1}]

z=c+r Exp[I t];
I2=Integrate[1/(z-c) D[z,t],{t,0,2Pi}]
I2a=Integrate[1/(z-c)^n D[z,t],{t,0,2Pi}]

z=(1+I)t;
I3=Integrate[Im[z] D[z,t],{t,0,1}]

z=Exp[I t];
I4=Integrate[Conjugate[z] D[z,t],{t,Pi,2Pi}]

z1=t;
z2=1+I*t;
z3=1+I-t;
z4=I-I*t;
I5=Integrate[Abs[z1]D[z1,t],{t,0,1}]+Integrate[Abs[z2]D[z2,t],{t,0,1}]
 +Integrate[Abs[z3]D[z1,t],{t,0,1}]+Integrate[Abs[z4]D[z2,t],{t,0,1}]

f[z_] := z^2 + 3 z + 4
I6 = Integrate[f[z1] D[z1, t], {t, 0, 1}] +
  Integrate[f[z2] D[z2, t], {t, 0, 1}] +
  Integrate[f[z3] D[z3, t], {t, 0, 1}] +
  Integrate[f[z4] D[z4, t], {t, 0, 1}]
```

答は (1) $I_1 = \frac{1}{2} + \frac{2}{3}i$ (2) $n=1$ のとき $I_2 = 2\pi i$, $n \neq 1$ のとき $I_2 = 0$ (3) $I_3 = \frac{1+i}{2}$ (4) $I_4 = \pi i$ (5) $I_5 = \frac{i-1}{2}\left(\sqrt{2} - 1 + \log(1+\sqrt{2})\right)$ (6) $I_6 = 0$

Mathematica は、(5) の $\log(1+\sqrt{2})$ を `ArcSinh[1]` と表示する。`TrigToExp[]` を施すと log で表示してくれる。

```
In[ ]:= TrigToExp[ArcSinh[1]]
Out[ ]= Log[1+√2]
```

11 曲線の連続的な変形 (ホモトピー)

円を楕円に変形する。

```
phi0[t_]:={Cos[t],Sin[t]};

phi1[t_]:={3Cos[t],2Sin[t]};

F[t_,u_]:=(1-u)phi0[t]+u*phi1[t];

Manipulate[ParametricPlot[{phi0[t], phi1[t], F[t, u]}, {t, 0, 2 Pi},
  PlotRange -> {{-3, 3}, {-3, 3}}], {u, 0, 1}]
```

phi1[t_]:={1/2,0} とすると、定数曲線で、像は $\{(1/2, 0)\}$ であり、円周 $x^2 + y^2 = 1$ を 1 点 $(1/2, 0)$ に変形することになる。

以上は複素数を使っていないので、書き換えてみる。

```
Clear[phi0,phi1,F]
phi0[t_]:=Exp[I t];
phi1[t_]:=3Cos[t]+2 I Sin[t];
F[t_,u_]:=(1-u)phi0[t]+u*phi1[t];
z2xy[z_]:={Re[z],Im[z]}
Manipulate[ParametricPlot[{z2xy[phi0[t]],z2xy[phi1[t]],z2xy[F[t,u]]},
                          {t, 0, 2 Pi}, PlotRange -> {{-3, 3}, {-3, 3}}],
          {u, 0, 1}]
```

参考文献

[1]